
Change 1
	Function name
	f_TC_13_3_2_2_EUTRA()

	Reason for change
	In the current TTCN, at step 13 UE is expected to send SNDCP data.

At step 12 there is handling for the UEs to perform XID negotiation for which there is a 10 seconds wait timer in the function f_GERAN_LLC_XID(), but it is not necessary for the UE to wait for 10 seconds before sending data.

As it is upto the UE to perform XID negotiation, uplink data should be handled if sent before 10 seconds.

In the CR R5s130368 above scenario was handled but only when UE sends SNDCP data after sending ‘two phase channel request’, 10s timer was stopped.

Similar handling should be done incase UE wants to send data after ‘one phase channel request’, but when UE is sending SNDCP data after sending ‘one phase channel request’ it is not possible to tell if this is for XID or SNDCP data, so we cannot stop the timer(10s) and exit the XID handling function at this point. 

Thus need to add handling for the SNDCP data inside the XID handling function. 

	Summary of change
	Introducing a new function f_GERAN_LLC_XID_Data() by merging two functions  f_GERAN_LLC_XID() and f_GERAN_Receive_UplinkSNDCP()

	TTCN module
	MultiLayer_Procedures_EG_GERAN.ttcn

	MCC160 Comment
	Accepted in principle but implemented differently, see below.  Simply check the reception of the SNDCP message before setting flag and ending the XID function.  Then when receiving the SNDCP data, only wait for a channel request if this flag is not set. 


Before change
	function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  { 

---

//@siclog "Step 10-11" siclog@

    //Send RAU Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType),

                                                                                    cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                    cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                                                    v_PDP_Context_Status)));

    //@siclog "Step 12-13" siclog@

    //Receive RAU Complete Message

    v_ChanReq := f_GERAN_LLC_XID (geran_Cell24, true); // @sic R5s130368 sic@

    //@siclog "Step 14" siclog@

    //Void

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    f_GERAN_Receive_UplinkSNDCP (geran_Cell24, v_ChanReq, car_G_SN_UnitData_IND (?, ?, v_IPv4IPv6_IcmpEchoReply)); // @sic R5s120570, R5s120861, R5s120919, R5s140108 sic@

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

 ---   

}


After change
	function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

{

---

//@siclog "Step 10-11" siclog@

    //Send RAU Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType),

                                                                                    cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                    cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                                                    v_PDP_Context_Status)));

    //@siclog "Step 12-13" siclog@

    //Receive RAU Complete Message

    v_ChanReq := f_GERAN_LLC_XID_Data (geran_Cell24, true);
    //@siclog "Step 14" siclog@

    //Void

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    //f_GERAN_Receive_UplinkSNDCP (geran_Cell24, v_ChanReq, car_G_SN_UnitData_IND (?, ?, v_IPv4IPv6_IcmpEchoReply)); // @sic R5s120570, R5s120861, R5s120919, R5s140108 sic@
    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

---

    }


New Function to be added:

	function f_GERAN_LLC_XID_Data(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete,

                           template LLCUnitDataIndType p_Expected_G_RA_UpdComplete := cr_G_RA_UpdComplete,

                           boolean p_WaitToSendResponse := false)

    runs on GERAN_PTC return template (omit) G_L2_DATAMESSAGE_IND // @sic R5s130368 sic@

  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get(p_CellId); // @sic R5s120587 sic@

    var template (omit) G_L2_DATAMESSAGE_IND v_ReturnChanReqTemplate := omit; // @sic R5s130368 sic@

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    timer t_Wait10S := 10.0;

    var octetstring v_IPv4IPv6_IcmpEchoReply := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE()); 

    var boolean p_SNDCPData := true;

        t_Wait10S.start;

    alt {

      []  G_LLC.receive(car_G_LLC_UnitData_IND(p_Expected_G_RA_UpdComplete)) // @sic R5s120837 sic@

        { 

               repeat;

        }

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        { 

          G_RLC.send(cas_G_RLC_ControlMsg_REQ_PACCH(p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment(v_BCCH_ARFCN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6)))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive(car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          G_CLLC.send(cas_G_CLLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, v_XID.xid.xID_Info, enum2int(p_CellId))); // @sic R5-135000 sic@

          if (p_WaitToSendResponse) { // @sic R5s141389 sic@

            f_Delay (3.0);

          }

          f_GPRS_SendDownlinkMsg(p_CellId, 1, cas_G_LLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, omit, enum2int(p_CellId)));

        }

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        { 

          repeat;

        }

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase))

        -> value v_ChanReq

        { 

          f_GERAN_ULTBFOnePhase(p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqTwoPhase)) // @sic R5s130368 sic@

        -> value v_ChanReq

        { 

          t_Wait10S.stop;

          v_ReturnChanReqTemplate := v_ChanReq;

        }

      [] G_SN.receive(?)

           {

             G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION at the end of the TBF

            p_SNDCPData := false;

           }

      [] t_Wait10S.timeout {} // No XID coming

    }

  if(p_SNDCPData)

    {

     f_GERAN_Receive_UplinkSNDCP (geran_Cell24, v_ReturnChanReqTemplate, car_G_SN_UnitData_IND (?, ?, v_IPv4IPv6_IcmpEchoReply)); // @sic R5s120570, R5s120861, R5s120919, R5s140108 sic@

    }

    return v_ReturnChanReqTemplate;  // @sic R5s130368 sic@

  }


MCC160 Implementation:
  function f_GERAN_Receive_UplinkSNDCP(GERAN_CellId_Type p_CellId,

                                       boolean p_TBF_AlreadyConfigured,
                                       template (present) G_SN_DATAMESSAGE_IND p_SNDCP_Ind := ?) runs on GERAN_PTC return G_SN_DATAMESSAGE_IND

  {

    var G_SN_DATAMESSAGE_IND v_RxdMsg;

    if (not p_TBF_AlreadyConfigured) { // @sic R5s150824 sic@

      f_GERAN_Receive_UplinkDataStart(p_CellId); // @sic R5s150326 sic@

    }
    // Receive the message

    interleave {

      [] G_SN.receive(p_SNDCP_Ind) -> value v_RxdMsg

        { }

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)) // RACE CONDITION at the end of the TBF

        {}

    }

    return v_RxdMsg;

  }

    function f_GERAN_LLC_XID(GERAN_CellId_Type p_CellId,

                           boolean p_RAUComplete,

                           template LLCUnitDataIndType p_Expected_G_RA_UpdComplete := cr_G_RA_UpdComplete,

                           boolean p_WaitToSendResponse := false,

                           boolean p_ExpectingSNDCP := false)

    runs on GERAN_PTC return boolean // @sic R5s130368, R5s150824 sic@

  {

    var B10_Type v_BCCH_ARFCN := f_GERAN_GPRSFreq_Get(p_CellId); // @sic R5s120587 sic@

 //   var template (omit) G_L2_DATAMESSAGE_IND v_ReturnChanReqTemplate := omit; // @sic R5s130368 sic@

    var G_L2_DATAMESSAGE_IND v_ChanReq;

    var G_LLC_DATAMESSAGE_IND v_XID;

    var boolean v_ReceivedSN := false; // @sic R5s150824 sic@

    timer t_Wait10S := 10.0;

    t_Wait10S.start;

    alt {

      [p_RAUComplete]  G_LLC.receive(car_G_LLC_UnitData_IND(p_Expected_G_RA_UpdComplete)) // @sic R5s120837 sic@

        {

          repeat;

        }

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketResourceRequestAny))

        {

          G_RLC.send(cas_G_RLC_ControlMsg_REQ_PACCH(p_CellId, tsc_PhyCh1, 1, tsc_UplinkTFI, cs_PacketUplinkAssignment(v_BCCH_ARFCN, substr(oct2bit(f_GERAN_TA_Get(p_CellId)),1, 6)))); // @sic R5s120587 sic@

          repeat;

        }

      [] G_LLC.receive(car_G_LLC_XID_IndAny)

        -> value v_XID

        {

          t_Wait10S.stop;

          // Wait for end of TBF

          // RACE CONDITION: Do we need to check the flag here ???

          G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement)); // RACE CONDITION

          G_CLLC.send(cas_G_CLLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, v_XID.xid.xID_Info, enum2int(p_CellId))); // @sic R5-135000 sic@

          if (p_WaitToSendResponse) { // @sic R5s141389 sic@

            f_Delay (3.0);

          }

          fl_SendXIDRes(p_CellId, cas_G_LLC_XID_Req(f_GERAN_TLLI_Get(p_CellId), v_XID.xid.sAPI, omit, enum2int(p_CellId))); // @sic R5s130195 Baseline Moving sic@

        }

      [] G_RLC.receive(car_G_RLC_ControlMsg_IND(p_CellId, tsc_PhyCh1, cr_PacketControlAcknowledgement))

        {

          // RACE CONDITION: Do we need to set a flag here ???

          repeat;

        }

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqOnePhase))

        -> value v_ChanReq

        {

          f_GERAN_ULTBFOnePhase(p_CellId, v_ChanReq.access); // @sic R5s120050 MCC160 implementation sic@

          repeat;

        }

 /*   The default behaviour will handle the two phase access request  

      [] G_L2.receive(car_G_L2_ACCESS_IND(p_CellId, tsc_PhyCh0, tsc_RACH, ?, ?, cr_ChanReqTwoPhase)) // @sic R5s130368 sic@

        -> value v_ChanReq

        {

          t_Wait10S.stop;

          v_ReturnChanReqTemplate := v_ChanReq;

        }*/
      [p_ExpectingSNDCP] G_SN.check(receive (?)) // @sic R5s150824 sic@

        {

          t_Wait10S.stop;

          v_ReceivedSN := true;

        }
      [] t_Wait10S.timeout {} // No XID coming

    }

    return v_ReceivedSN;  // @sic R5s130368 sic@

  } //end of f_GERAN_LLC_XID

    function f_TC_13_3_2_2_GERAN () runs on GERAN_PTC

  { /* Inter-system connection re-establishment / E-UTRAN to GPRS / Further data are to be transferred */

    var G_LLC_DATAMESSAGE_IND v_LLCDataInd;

    var B3_Type v_UpdateType;

    var PDP_ContextStatus v_PDP_Context_Status;

    var GERAN_SecurityInfo_Type v_SecurityInfo;

    var octetstring v_PLMN;

    var O1_Type v_RAC;

    var O2_Type v_LAC;

    var boolean p_TBF_AlreadyConfigured := false; // @sic R5s130368, R5s150824 sic@

    var octetstring v_IPv4IPv6_IcmpEchoReply := f_IPv4IPv6_IcmpEchoReply(f_LoopbackModeB_IP_Address_UE());      /* @sic R5s120470 sic@ */

    //Initialise test case

    f_GERAN_Init(EUTRA_GERAN);

    f_GERAN_CreateCell (geran_Cell24);

    v_PLMN := f_GERAN_PLMN_Get(geran_Cell24);

    v_RAC := f_GERAN_RAC_Get(geran_Cell24);

    v_LAC := f_GERAN_LAC_Get(geran_Cell24);

    //Receive auth parameters and set cell power level of cell 24 to suitable according to TS 36.508 subclause 6.2.2.1

    f_GERAN_InterRAT_InitialiseAuthParams();

    f_GERAN_SetCellPowerLevel (geran_Cell24, tsc_GERAN_SuitableNeighbourCell);

    f_GERAN_TestBody_Set(true);

    //Create the physical channel in GERAN cell ?

    //f_GERAN_CreateTCH (geran_Cell24);

    //@siclog "Step 2" siclog@

    //@siclog "Step 3-5" siclog@

    //Receive RAU Request Message

    v_LLCDataInd := f_GPRS_ReceiveUplinkMsg (geran_Cell24,

                                             car_G_LLC_UnitData_IND (cr_G_RA_UpdReqAny (cr_GMM_UpdateType('0'B, (tsc_I_UpdateTypeRAOnly,

                                                                                                                 tsc_I_UpdateTypeCombined,

                                                                                                                 tsc_I_UpdateTypeCombinedWithIMSI))))); // @sic R5-125545 sic@

    f_GERAN_SNDCP_Config (1, 1, 5, 11); // @sic R5s12861, R5s130368, R5s130883 sic@

    //Check PDP context status (shall be indicated as not PDP-INACTIVE)

    v_PDP_Context_Status := v_LLCDataInd.unit.msg.rauReq.pDP_ContextStatus;

    if (match(v_PDP_Context_Status, cs_PDP_ContextStatusInactive)){

      f_GERAN_SetVerdictFailOrInconc(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3 - PDP Context Status is incorrect/PDP-inactive");

    }

    else {

      f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 3");

    }

    v_UpdateType := f_UG_RauReq_GetUpdateValueForRauAcc(v_LLCDataInd.unit.msg.rauReq);

    if (v_UpdateType == tsc_I_UpdateTypeCombinedWithIMSI or v_UpdateType == tsc_I_UpdateTypeCombined) {

      v_SecurityInfo := f_GERAN_SecurityInfo_Get();

      v_SecurityInfo.CSinOtherRAT := true;

      f_GERAN_SecurityInfo_Set(v_SecurityInfo);

    }

    //@siclog "Step 6-9" siclog@

    //Perform Authentication

    f_GERAN_GPRS_Authentication (geran_Cell24);

    //@siclog "Step 10-11" siclog@

    //Send RAU Accept

    f_GPRS_SendDownlinkMsg(geran_Cell24, 1, cas_G_LLC_UnitData_Req (f_GERAN_TLLI_Get(geran_Cell24),

                                                                    tsc_LLCSapi_GMM,

                                                                    tsc_LLC_PM,

                                                                    px_GSM_CipheringOnOff,

                                                                    cs_G_RA_UpdAcc3(cs_GMM_UpdateResult(v_UpdateType),

                                                                                    cs_RAI(omit, cs_LAI(omit, v_PLMN, v_LAC), v_RAC),

                                                                                    cs_PTMSI_Signature (px_PTMSI_SigDef),

                                                                                    cs_MobileIdTMSIorPTMSI ('18'O, px_PTMSI_Def),

                                                                                    v_PDP_Context_Status)));

    //@siclog "Step 12-13" siclog@

    //Receive RAU Complete Message

    p_TBF_AlreadyConfigured := f_GERAN_LLC_XID (geran_Cell24, true, -, -, true); // @sic R5s130368, R5s150824 sic@

    //@siclog "Step 14" siclog@

    //Void

    //@siclog "Step 15" siclog@

    //Check: Does the UE loop back the IP packet on the DRB associated with the default EPS bearer context? */

    f_GERAN_Receive_UplinkSNDCP (geran_Cell24, p_TBF_AlreadyConfigured, car_G_SN_UnitData_IND (?, ?, v_IPv4IPv6_IcmpEchoReply)); // @sic R5s120570, R5s120861, R5s120919, R5s140108, R5s150824 sic@

    f_GERAN_PreliminaryPass(__FILE__, __LINE__, "Test case 13.3.2.2 Step 15");

    //Tell to EUTRA PTC Test case is now finished on GERAN side

    f_IRAT_SendCoOrd (EUTRA, cms_IRAT_Trigger);

    f_GERAN_TestBody_Set(false);

    //Switch/power off UE and release cell

    f_GERAN_Postamble (geran_Cell24, G1_IDLE);

  }

